
Введение

Решение задач линейной алгебры – одна из основных задач высшей математики и экономики. В наше время трудно представить мир без линейной алгебры. Очень часто задачи линейной алгебры решают такие проблемы как: сжатие данных, прогноз погоды, аналитический, экономические прогнозы и мн. др.

Целью данной работы является изучение основных «точных» методов решения задач линейной алгебры уравнений. Таких как решения систем линейных алгебраических уравнений методом Гаусса с выбором главного элемента. Изучен алгоритм произведения матриц. В рамках поставленной цели был рассмотрен теоретический материал и составлены программы на языке Free Pascal. Поэтому работа состоит их двух частей (методов): теоретической и практической. При решении задачи учитывались возможные «крайние» случаи СЛАУ – вырожденность или плохая обусловленность системы. Был разобран тестовый пример системы и произведена проверка решения. 

Данную работу можно считать актуальной поскольку, область практического применения её результатов, как и большинства прикладных математических методов, распространяется на все области человеческой деятельности, где используется математическое моделирование.

На сегодня главной проблемой решения задач линейной алгебры является временные и пространственные затраты. Поэтому оптимизация решения с помощью новейших технологий распараллеливания будет актуальна еще долгое время. 

Предполагается использование средств распараллеливания и к другим методам курса высшей математики. 
1 Установка компилятора Free Pascal и MPI Chamelleon
Для начала установки нужно приобрести установочныый пакет рабочей (тестированной на платформе Windows) версии компилятора.

При установке Free Pascal особых изменений вводить не нужно (по желанию).

MPICH для Windows требует:

Windows NT4/2000/XP (Professional или Server). Не поддерживает Win9x/ME! 

Сетевое соединение по протоколу TCP/IP между машинами. 

Компьютеры, участвующие в вычислениях, назовем кластером. MPICH должен быть установлен на каждом компьютере в кластере. 

Стандартные настройки MPICH можно оставить как есть (Установить по умолчанию)

Настройка MPI Chameleon
Настройку можно осуществить с помощью простых утилит, имеющихся в дистрибутиве. Для этого нужно:

Запустить MPIConfig.exe (можно воспользоваться ссылкой в главном меню, она там должна быть) 

Нажать на кнопку "Select" 

В появившемся окне выбрать пункт меню "Action"—"Scan hosts" 

Напротив имени каждой машины должна загореться пиктограмма "MPI" (как на рисунке ниже)

После этого можно запускать (с помощью загрузчика) MPI-приложения. Но для того чтобы их отладить нужно интегрировать MPICH в Free Pascal.

Теперь чтобы «привязать» библиотеки MPICH к Free Pascal следует воспользоваться динамической библиотекой mpich.dll, которая располагается в системном каталоге (копируется туда при установке MPICH).

Для этого:

Приобрести модуль FreePascal, реализующий функции этой динамической библиотеки. Файл mpi.pp.

Для использования модуля mpi следует просто скопировать файл mpi.pp в каталог, где FreePascal ищет модули (unit searchpath). По умолчанию это каталог «units\ i386-win32\» в корневой папке программы.

Все вышеуказанные ПО присутствуют на данном диске в папке Soft. Также на диске в папках Generator и Launcher есть генератор СЛАУ для последующего решения и проверки и программа для облегчения запуска MPI приложений (Визуальное управление загрузчиком MPIRun.exe)


2 Постановка задач

Требуется найти произведение двух матриц размерностью Аm×l  и Вl×n
В случае отсутствия во входном файле значений элементов матриц заполнить матрицы случайными числами

Требуется решить систему линейных алгебраических уравнений с вещественными коэффициентами вида

a11x1 + a12x2 + … + a1nxn = b1 ,
a21x2 + a22x2 + … + a2nxn = b2 ,
.   .   .   .   .   .   .   .   .   .   .   .   .

an1x1 + an2x2 + … + annxn = bn
по методу Гаусса.

Проверить результаты решения системы линейных уравнений

2.1 Тестовый примеры

Решение первой задачи

       2.67      7.32      3.93      8.71

A = 3.85      1.04      0.96      0.72

       1.44      0.97      1.41      2.22

       9.71      8.59      8.77      9.79      7.01

B = 5.49      7.92      5.16      8.63      7.15

       8.34      8.85      4.65      5.26      4.20

       6.62      8.36      8.55      3.45      8.74

      156.53    188.55    153.91    140.00    163.67

C = 55.91     55.89     49.80     54.24     44.79

       45.79     51.14     43.22     37.59     42.40

Решение СЛАУ [3-6]:

3,2x1 + 5,4x2 + 4,2x3 + 2,2x4 = 2,6 ,

2,1x1 + 3,2x2 + 3,1x3 + 1,1x4 = 4,8 ,

1,2x1 + 0,4x2 – 0,8x3 – 0,8x4 = 3,6 ,

4,7x1 + 10,4x2 + 9,7x3 + 9,7x4 = –8,4 ,

x1 = 5, x2 = –4, x3 = 3, x4 = –2.

Проверка:

3,2*5 + 5,4*(–4) + 4,2*3 + 2,2*(–2) = 2,6 ,

2,1*5 + 3,2*(–4) + 3,1*3 + 1,1*(–2) = 4,8 ,

1,2*5 + 0,4*(–4) – 0,8*3 – 0,8*(–2) = 3,6 ,

4,7*5 + 10,4*(–4) + 9,7*3 + 9,7*(–2) = –8,4 ,

2.2 Описания и листинги программ 

I программа. В данной программе реализовано произведение матриц. Данные вводятся с помощью файлов имена которых задаются с помощью параметров командной строки (1 параметр – входной, 2-ой – выходной файлы). Т. е.:

Имя_программы Имя_входного_файла Имя_выходного_файла (PGAUSS.EXE input.txt output.txt)

Параметры MPIRun.exe вводятся отдельно до названия программы.

На первой строке входного файла указывается размерность матриц (N, L, M). Далее через разделитель записываются элементы матриц. Пример: 


3
4
5

      2.67      7.32      3.93      8.71

      3.85      1.04      0.96      0.72

      1.44      0.97      1.41      2.22
      9.71      8.59      8.77      9.79      7.01

      5.49      7.92      5.16      8.63      7.15

      8.34      8.85      4.65      5.26      4.20

      6.62      8.36      8.55      3.45      8.74

В выходной файл программа записывает значения матриц А и Б, время вычислений и значения матрицы С.

Листинг
program UMParN;  

uses Mpi;

const

   MaxM = 8;  MaxL = 8;  MaxN = 8;

type

   Float = Double;

var

   F, G: Text;

   I, J, K, M, L, N: Integer;

   A: array[1..MaxM, 1..MaxL] of Float;

   B: array[1..MaxL, 1..MaxN] of Float;

   C: array[1..MaxM, 1..MaxN] of Float;

   NumProcs, MyId, Tag: LongInt;

   Status: MPI_Status;

   StartTime, EndTime: Double;

begin

   MPI_Init(Argc, Argv);  Tag := 0;

   MPI_Comm_size(MPI_COMM_WORLD, NumProcs);

   MPI_Comm_rank(MPI_COMM_WORLD, MyId);

   if MyId = 0 then begin
        {Имена входного и результирующего файлов - это 1-й и 2-й

         параметр командной строки}

      if ParamCount <> 2 then begin

         WriteLn('Bad arguments!');

         Halt(1)

      end;
      Assign(F, ParamStr(1));  Reset(F);

      Assign(G, ParamStr(2));  Rewrite(G);

        {Ввод реальных размеров матриц: M, L и N}

      ReadLn(F, M, L, N);

      if Eof(F) then begin
          {Генерация матриц A и B случайным образом}

         Randomize;

         WriteLn(G, 'A =');

         for I := 1 to M do begin

            for J := 1 to L do begin

               A[I, J] := 10 * Random;  Write(G, '  ', A[I, J]: 8: 2)

            end;
      {Строки A сразу рассылаем по соответствующим процессам}

            K := (I - 1) mod NumProcs;

            if K > 0 then

               MPI_Send(@A[I], L, MPI_DOUBLE, K, Tag,

       MPI_COMM_WORLD);

            WriteLn(G)

         end;
         WriteLn(G, 'B =');

         for I := 1 to L do begin

            for J := 1 to N do begin

               B[I, J] := 10 * Random;  Write(G, '  ', B[I, J]: 8: 2)

            end;
            WriteLn(G)

         end

      end

      else begin

          {Ввод матриц A и B из файла}

         for I := 1 to M do begin

            for J := 1 to L do

               Read(F, A[I, J]);

      {Строки A сразу рассылаем по соответствующим процессам}

            K := (I - 1) mod NumProcs;

            if K > 0 then

               MPI_Send(@A[I], L, MPI_DOUBLE, K, Tag,

       MPI_COMM_WORLD)

         end;
         for I := 1 to L do

            for J := 1 to N do

               Read(F, B[I, J])

      end;
      StartTime := MPI_Wtime
   end;      {Коллективная рассылка размеров матриц}

   MPI_Bcast(@M, 1, MPI_INT, 0, MPI_COMM_WORLD);

   MPI_Bcast(@L, 1, MPI_INT, 0, MPI_COMM_WORLD);

   MPI_Bcast(@N, 1, MPI_INT, 0, MPI_COMM_WORLD);

     {Коллективная рассылка (построчная!) матрицы B}

   for I := 1 to L do

      MPI_Bcast(@B[I], N, MPI_DOUBLE, 0,

MPI_COMM_WORLD);

{C = A * B (каждый процесс считает свою горизонтальную полосу)}

   I := MyId + 1;

   if MyId = 0 then       {Root-процесс считает как обычно}

      while I <= M do begin

         for J := 1 to N do begin

            C[I, J] := 0;

            for K := 1 to L do

               C[I, J] := C[I, J] + A[I, K] * B[K, J]

         end;
         I := I + NumProcs

      end

   else                   {Не root-процесс должен сначала получить A[I]}

      while I <= M do begin

         MPI_Recv(@A[I], L, MPI_DOUBLE, 0, Tag,

  MPI_COMM_WORLD, Status);

         for J := 1 to N do begin

            C[I, J] := 0;

            for K := 1 to L do

               C[I, J] := C[I, J] + A[I, K] * B[K, J]

         end;             {а затем обратно отослать строку C[I]}

         MPI_Send(@C[I], N, MPI_DOUBLE, 0, Tag,

  MPI_COMM_WORLD);

         I := I + NumProcs

      end;
   if MyId = 0 then begin

      EndTime := MPI_Wtime;

      WriteLn(G, 'Parallel computations: ',

                     EndTime - StartTime, ' sec.');

      WriteLn(G, 'C = A * B =');

      for I := 1 to M do begin

         K := (I - 1) mod NumProcs;

         if K > 0 then    {Перед выводом надо получить C[I]}

            MPI_Recv(@C[I], N, MPI_DOUBLE, K,

                               Tag, MPI_COMM_WORLD, Status);

         for J := 1 to N do

            Write(G, '  ', C[I, J]: 8: 2);

         WriteLn(G)

      end;
      Close(G)

   end;
   MPI_Finalize

end.

II программа. В данной программе реализован метод Гаусса со схемой частичного выбора. 

Чтобы ввести данные из файла нужно воспользоваться командной строкой с двумя параметрами (соответственно, имена входного и выходного файлов). Пример использования:

Имя_программы Имя_входного_файла Имя_выходного_файла (PGAUSS.EXE input.txt output.txt)

Параметры MPIRun.exe вводятся отдельно до названия программы.

На первой строке входного файла указывается размерность матрицы (n), а на последующих n строках через разделитель записываются элементы матрицы a и вектора b. Пример: 

2

2
3

4

3
4

5

В выходной файл программа записывает результаты вычислений. На первой строке файла записывается строка “Значения x:”, в последующих n (в зависимости от числа элементов вектора x) строках значения вектора x. После, идет проверка, пишется строка “Проверка”, на следующих n сумма всех произведений коэффициентов a1i,j на x j и правая сторона системы (b1i). Пример:

Значения x:

    x1 = -1.0000000000E+00

    x2 =  2.0000000000E+00

Проверка

a1*x =  4.0000000000E+00 b1 =  4.0000000000E+00

a1*x =  5.0000000000E+00 b1 =  5.0000000000E+00

В случае плохой обусловленности вывести единственную строку “Данная система плохо обусловлена!”. Пример:

Данная система плохо обусловлена!

Листинг

program PGauss;

Uses CRT, MPI;

Const

  maxn = 6;

  eps = 1E-6;

Type

  Data = Double;

  Matrix = Array[1..10000, 1..10000] of Data;

  Vector = Array[1..10000] of Data;

{ Объявление переменных }

Var
  Tag, NumProcs, MyID : Longint; //Тег сообщения, количество процессов, номер процесса(curent)

  StartTime, EndTime: Double;    //Время начала и конца вычислений

  n, i, j, k: Integer;           //Счетчики циклов

  a, a1: Matrix;                 //Матрица a и временная матрица a1

  b, b1, x: Vector;              //Вектор b, временный вектор b1 и вектор x
  Status: MPI_Status;            //Статус принятого сообщения

  BadCode: Byte;                 //Код, определяющий тип плохой обусловленности

  F: Text;                       //Файловая переменная

{ Описание процедур }

{ Процедура ввода расширенной матрицы системы }

Procedure ReadSystem(n: Integer; var a: Matrix; var b: Vector);

const

  znaki = 10;

Var

  i, j, r: Integer;

Begin

  r := WhereY;

  GotoXY(2, r);

  Write('A');

  For i := 1 to n do begin

    GotoXY(i * znaki - znaki + 6, r);

    Write(i);

    GotoXY(1, r + i + 1);

    Write(i: 2);

  end;

  GotoXY((n + 1) * znaki - znaki + 6, r);

  Write('b');

  TextColor(LightCyan);

  For i := 1 to n do begin

    For j := 1 to n do begin

      GotoXY(j * znaki - znaki + 6, r + i + 1);

      Read(a[i, j]);

    end;

    GotoXY((n + 1) * znaki - znaki + 6, r + i + 1);

    Read(b[i]);

  end;

  if NumProcs <> 1 then

    MPI_BCast(@n, 1, MPI_Integer, 0, MPI_COMM_WORLD);

end;

{ Процедура ввода расширенной матрицы системы с файла }

Procedure ReadSystemF(var n: Integer; var F: text; var a: Matrix; var b: Vector);

const

  znaki = 10;

Var

  i, j: Integer;

Begin

  ReadLn(F, n);

  For i := 1 to n do begin

    For j := 1 to n do begin

      Read(F, a[i, j]);

    end;

    Read(F, b[i]);

  end;

  if NumProcs <> 1 then

    MPI_BCast(@n, 1, MPI_Integer, 0, MPI_COMM_WORLD);

end;

{ Процедура вывода значений x и результатов проверки }

Procedure WriteX(n :Integer; const a1: Matrix;

                 const b1: Vector; const x: Vector);

Var

  i, j: Integer;

  s: Data;

Begin

  TextColor(Yellow);

  WriteLn('******************************************************************');

  WriteLn('Результат вычислений по методу Гаусса');

  For i := 1 to n do

    Writeln('                                      x', i, ' = ', x[i]);

  WriteLn('******************************************************************');

  WriteLn;

  WriteLn;

  {***Проверка***}

  TextColor(White);

  WriteLn('Проверка');

  for i := 1 to n do begin

    s := 0;

    for j := 1 to n do

      s := s + a1[i, j] * x[j];

    WriteLn('a1 * x = ', s, '        b1 = ', b1[i])

  end
End;

{ Процедура вывода значений x и результатов проверки в файл }

Procedure WriteXF(n :Integer; var F: text; const a1: Matrix;

                 const b1: Vector; const x: Vector);

Var

  i, j: Integer;

  s: Data;

Begin

  WriteLn(F, 'Значения x:');

  For i := 1 to n do

    WriteLn(F, '    x', i, ' = ', x[i]);

  {***Проверка***}

  WriteLn(F);

  WriteLn(F, 'Проверка');

  for i := 1 to n do begin

    s := 0;

    for j := 1 to n do

      s := s + a1[i, j]*x[j];

    WriteLn(F, 'a1 * x = ', s, ' b1 = ', b1[i])

  end
End;

{ Процедура, реализующая обратный ход }

procedure Reverse(n: Integer; var a: Matrix; var b: Vector; var x:Vector);

Var

  i, j, k, l: Integer;

  w: Data;

begin
  { Вычисляем решение (обратный ход) }

  if (MyID = 0) and (NumProcs <> 1) then

  begin

    For i := n downto 1  do

    begin

      k := (i - 1) mod NumProcs;

      if k > 0 then

      begin
        { Рассылка строк матрицы a по процессам }

        MPI_Send(@a[i], n, MPI_Double, k, Tag, MPI_COMM_WORLD);

      end;

    end;

  end;

  For i := n downto 1 do

  begin

    k := (i - 1) mod NumProcs;

    if k = MyID then

      begin

        w := 0;

        For j := i + 1 to n do

          w := w + a[i, j] * x[j];

        x[i] := (b[i] - w) / a[i, i];

      end;

    if NumProcs <> 1 then
      { Коллективная рассылка значений вектора x }

      MPI_BCast(@x[i], 1, MPI_Double, k, MPI_COMM_WORLD);

  end
end;

{ Функция, реализующая метод Гаусса }

Function Gauss(n: Integer; var a: Matrix; var b: Vector; var BadCode: Byte): Boolean;

Var

  i, j, k, l: Integer;

  w: Data;

Begin
  { Контрольная точка начала математических вычислений }

  StartTime := MPI_WTime;

  { Приводим матрицу к верхнему треугольному виду (прямой ход) }

  For k := 1 to n do

  begin

    { Ищем строку l с максимальным элементом в k-ом столбце}

    w := 0;

    l := 0;

    For i := k to n do

    begin

      If Abs(a[i, k]) > w then

        w := Abs(a[i, k]);

      l := i;

    end;

    { Меняем местом l-ую строку с k-ой }

    If l <> k then begin

      For j := k to n do begin

        w := a[k, j];

        a[k, j] := a[l, j];

        a[l, j] := w;

      end;

      w := b[k];

      b[k] := b[l];

      b[l] := w;

    end;

    { Если главный элемент мал, то система плохо обусловлена }

    If Abs(a[k, k]) < eps then

    begin

      Gauss := false;

      if (a[k, k] = 0) and (b[k] = 0)   // 1) Система имеет бесконечное

        then BadCode := 1 else          //    множество решений;

      if (a[k, k] = 0) and (b[k] <> 0)  // 2) Система не имеет решений;

        then BadCode := 2 else          // 3) Система почти вырождена.

        BadCode := 3;

      Exit; //Вывод ошибки и завершение работы программы

    end;

    { Преобразуем матрицу }

    For i := k + 1 to n do

    begin

      w := a[i, k] / a[k, k];

      For j := k + 1 to n do

        a[i, j] := a[i, j] - w * a[k, j];

      b[i] := b[i] - w * b[k];

    end;

  end; {for k}

  if NumProcs <> 1 then

    MPI_BCast(@b, n, MPI_Double, 0, MPI_COMM_WORLD);

  Reverse(n, a, b, x);

  Gauss := True;

End;

{ Тело программы }

Begin
{ Инициализация программы }

  ClrScr;

  TextColor(White);

  BadCode := 0;

{ Инициализация MPI }

  MPI_Init(ArgC, ArgV);

  Tag := 0;

  MPI_Comm_Size(MPI_COMM_WORLD, NumProcs);

  MPI_Comm_Rank(MPI_COMM_WORLD, MyID);

{ Эту часть алгоритма выполняет только root процесс }

(* Начало *)

  if MyID = 0 then

  begin

    WriteLn(' ------------------------------------------------------------ ');

    TextColor(LightGray);

    Writeln('|Программа решения систем линейных уравнений по методу Гаусса|');

    TextColor(White);

    WriteLn(' ------------------------------------------------------------ ');

    Writeln('                                        Запущено процессов: ', NumProcs);

    Writeln;

    Writeln;

{ Файловый обмен с диском }

    if ParamCount = 2 then

    begin

      Assign(F, ParamStr(1)); Reset(F);

        { Ввод из файла }

        TextColor(Yellow);

        Writeln('               <<<Режим работы с файлами>>>');

        WriteLn('                  ----------------------   ');

        TextColor(LightGray);

        Write('        Ввод: ');

        TextColor(White);

        WriteLn(ParamStr(1));

        TextColor(LightGray);

        Write('        Вывод: ');

        TextColor(White);

        WriteLn(ParamStr(2));

        ReadSystemF(n, F, a, b);

        for i := 1 to n do

        begin

          b1 := b;

          for j := 1 to n do

            a1[i, j] := a[i, j];

        end;

      Close(F);

      Assign(F, ParamStr(2)); Rewrite(F);

        { Вывод в файл }

        If Gauss(n, a, b, BadCode) then

          WriteXF(n, F, a1, b1, x)

        else

        begin

          TextColor(LightRed);

          Writeln(F, 'Данная система плохо обусловлена!');

          case BadCode of
            1: WriteLn(F, 'Система уравнений имеет бесконечное множество решений');

            2: WriteLn(F, 'Система уравнений не имеет решений');

            3: WriteLn(F, 'Система уравнений является почти вырожденной');

          end;

        end;

      Close(F)

    end

    else

{ Ввод/вывод данных и результатов с клавиатуры и на экран }

    begin
      Repeat
        Writeln('Введите порядок матрицы системы (макс. ', maxn, ')');

        Write('>');

        Read(n);

        if n <= 0 then

        begin

          TextColor(LightRed);

          WriteLn('Порядок матрицы системы должен состоять только из положительных чисел!');

          TextColor(White);

        end;

        if n > maxn then

        begin
          TextColor(LightRed);

          WriteLn('Порядок матрицы системы не должен превышать числа ',  maxn, '');

          TextColor(White);

        end;

        WriteLn;

      Until (n > 0) and (n <= maxn);

      Writeln;

      Writeln('Введите матрицу системы и элементы правой части');

      ReadSystem(n, a, b);

      for i := 1 to n do

      begin

        b1 := b;

        for j := 1 to n do

          a1[i, j] := a[i, j];

      end;

      Writeln;

      If Gauss(n, a, b, BadCode) then

      begin

        TextColor(White);

        WriteX(n, a1, b1, x)

      end else

      begin

        TextColor(LightRed);

        Writeln('Данная система плохо обусловлена!');

        case BadCode of
          1: WriteLn('(Система уравнений имеет бесконечное множество решений)');

          2: WriteLn('(Система уравнений не имеет решений)');

          3: WriteLn('(Система уравнений является почти вырожденной)');

        end;

      end;

    end;

    Writeln;

    TextColor(LightMagenta);

    Writeln('       <<<Работа закончена. Нажмите любую клавишу>>>');

    EndTime := MPI_WTime;

    WriteLn;

    WriteLn;

    WriteLn;

    TextColor(LightGreen);

    WriteLn('       Потрачено времени на вычисления: ', (EndTime - StartTime): 0: 5, ' sec.');

  end;

(* Конец *)

{ Этот отрывок выполняется только дочерними процессами }

(* Начало *)

  if MyID <> 0 then
  begin
    { Получение данных с root процесса }

    MPI_BCast(@n, 1, MPI_Integer, 0, MPI_COMM_WORLD);

    MPI_BCast(@b, n, MPI_Double, 0, MPI_COMM_WORLD);

    for i := 1 to n do

    begin

      k := (i - 1) mod NumProcs;

      if k = MyID then

        MPI_Recv(@a[i], n, MPI_Double, 0, Tag, MPI_COMM_WORLD, Status);

    end;

    Reverse(n, a, b, x);

  end;

(* Конец *)

  if MyID = 0 then ReadKey; //Ожидание нажатия клавиши.

{ Завершение работы MPI }

  MPI_Finalize;

{ Завершение работы программы }

end.

2.1.6 Результат выполнения программ
Скриншоты интерфейсов программ с использованием командной строки и без приведены на рисунках 5, 6 и 7. 

Рисунок 5. Программа решения систем линейных алгебраических уравнений методом Гаусса, запуск с помощью командной строки Windows (работа с файлами).

Рисунок 6. Программа решения систем линейных алгебраических уравнений методом Гаусса, запуск без использования командной строки Windows (ввод данных вручную).

Заключение
В данной работе рассмотрено решение основных задач линейной алгебры (решение систем линейных алгебраических уравнений методом Гаусса с выбором главного элемента и произведение матриц). Разобраны и учтены «крайние» случаи: вырожденность или плохая обусловленность матрицы системы. Составлены программы на языке Free Pascal, пропущены тестовые примеры, произведена проверка полученных результатов. В работе приводится листинг программ и картинка экрана с отображением полученных результатов. Программы могут организовать ввод/вывод данных и результатов с клавиатуры и на экран, а также реализовать файловый обмен с диском.

Результаты работы программы подтверждают правильность метода Гаусса и других основных задач линейной алгебры, возможность их использования в тех случаях математического моделирования, когда встаёт вопрос решения задач линейной алгебры. Проблемой при его использовании могут быть значительные временные затраты при больших матрице и векторе правой части. В этом случае методы линейной алгебры хорошо поддаются распараллеливанию вычислений.

Постоянный рост сложности и объёмов расчётов предъявляет повышенные требования к производительности компьютеров. Принципиально каждый компьютер состоит из каналов, переключателей и запоминающих элементов. Тогда его быстродействие пропорционально S / r + V, где S – скорость распространения сигналов; V – скорость переключения; r – среднее расстояние между элементами. Отсюда универсальная рекомендация для увеличения быстродействия – делать элементы ЭВМ, обладающие более высокой скоростью и меньшими размерами. Однако скорость распространения сигнала ограничена скоростью света, а слишком близкое "укладывание" элементов приводит к чрезмерному перегреванию. На текущем этапе увеличение быстродействия за счёт этих трёх характеристик в значительной степени исчерпало себя. Поэтому параллельные вычисления – это другой путь повышения производительности ЭВМ.

